Supplemental Digital Content 3: NONMEM Control File for Final Labor Pain Model

$PROB Final Model

$DATA allpain.data.csv

$INPUT ID RTIM DIL RATE GRP OXY EPI ROM IND NRS=DV NR

$PRED

    ;median rate = 0.372

    DR = RATE-NR

    E0TV   = THETA(1)    ; Baseline

    EMAXTV = THETA(2)    ; Max Effect

    C50TV  = THETA(3)    ; C50

    ; Calculate Gamma

    IF (GRP.EQ.1) THEN

       GAMTV  = THETA(4) + (THETA(6)*DR)  ; Asians

    ELSE

       GAMTV  = THETA(5) + (THETA(6)*DR)  ; Everyone else

    ENDIF

    E0   = E0TV   + ETA(1)

    EMAX = EMAXTV + ETA(2)

    C50  = C50TV  * EXP(ETA(3))

    GAM  = GAMTV

    TIME = DIL

    IPRD=E0+(EMAX-E0)*DIL**GAM/(DIL**GAM+C50**GAM)

    IF (IPRD.LT.0) IPRD = 0

    Y=IPRD+EPS(1)

$THETA 

   (0,  1.02, 10)    ; E0 (units = NRS) 

   (0,  8.73, 10)    ; Emax (units = NRS) 

   (0,  2.27, 10)    ; C50 (units = cm) 

   (.1, 2.19, 10)    ; Gamma SLOPE Asians

   (.1, 4.76, 10)    ; Gamma SLOPE Everyone Else

        2.26         ; Rate factor on E0

$OMEGA

   .245     ; Intersubject variance on E0

   0 FIX    ; Intersubject variance on Emax

   .311     ; Intersubject variance on C50

$SIGMA

   2.73     ; Residual noise

$ESTIMATION SIG=3 MAX=3000 PRINT=1  NOABORT METHOD = 1; FOCE method 

$TABLE ID RTIM TIME IPRD E0 EMAX C50 GAM OXY EPI ROM IND RATE GRP

FILE=ALLRECORDS.TXT ONEHEADER NOPRINT

